

Eclipses de Lune

Ephémérides astronomiques

Serveur de l'IMCCE

Mouvement de la Lune

La position du plan de l'orbite de la Lune inclinée de $5^{\circ}15'$ sur l'écliptique, et les variations rapides des éléments de celle-ci (nœuds, périégée...) font qu'au cours de ses révolutions successives, sa trajectoire apparente sur le ciel se modifie sans cesse. Assez régulièrement, la Lune passe dans le plan de l'écliptique et se trouve alignée avec l'axe Terre-Soleil. On a alors soit éclipse de Soleil, soit éclipse de Lune suivant les positions relatives des trois corps.


Grâce aux éphémérides précises géocentriques de l'IMCCE, nous allons pouvoir suivre le mouvement apparent de la Lune par rapport à la direction du Soleil, ou pour ce qui nous intéresse la direction opposée, l'axe du cône d'ombre de la Terre, et prédire avec une bonne précision les positions et instants des éclipses de Lune. Pour l'application concrète, nous allons étudier l'éclipse de Lune du 3 mars 2007.

Ephémérides

Nous nous plaçons dans le repère géocentrique.

Les éphémérides donnent à tout instant choisi :

- la direction du Soleil, opposée à la direction de l'axe du cône d'ombre L_S et l_S (longitude, latitude écliptiques)
- la direction de la Lune L_L et l_L
- la distance Terre Soleil D_S
- la distance Terre Lune D_L



Nous connaissons :

- le diamètre du Soleil R_S
- le diamètre de la Terre R_T
- le diamètre de la Lune R_L

Scénario

Pour suivre l'éclipse, nous allons calculer avec les éphémérides, pas à pas :

- θ l'angle entre la direction du centre de la Lune et la direction du cône d'ombre
- ρ_L le rayon angulaire de la Lune
- ρ_{ombre} le rayon angulaire du cône d'ombre à la hauteur de la Lune

Et tracerons les graphes des variations de ces grandeurs en fonction du temps

A partir des données, nous chercherons quand

- le centre de la Lune est au plus près du centre du cône d'ombre (maximum et grandeur de l'éclipse)
- le centre de la Lune est à $\rho_{\text{ombre}} + \rho_L$ du centre du cône d'ombre (entrée et sortie du cône d'ombre)
- le centre de la Lune est à $\rho_{\text{ombre}} - \rho_L$ du centre du cône d'ombre (début et fin de l'éclipse totale)

On ne s'occupera que de l'éclipse dans l'ombre de la Terre, et non de la pénombre.

Plan de travail

- 1) Téléchargement du site de l'IMCCE des éphémérides de la Lune et du Soleil
- 2) Transformation des données :
 - dates, heures
 - coordonnées écliptiques
 - distances
- 3) Calculs :
 - différences de longitudes et latitudes, distance angulaire cône d'ombre - Lune
 - Rayon angulaire de la Lune
 - Rayon du cône d'ombre
- 4) Tracé des graphiques
- 5) Calcul des instants de l'éclipse
 - approximation, interpolation, ajustement
 - Recherche des instants du maximum, des contacts.
- 6) Approfondissement
 - Retour sur le calcul des instants avec un peu plus de mathématiques pour plus de rigueur.
- 7) Résultats

Fichier tableur et feuilles de calcul

Le stockage et la manipulation des données se fait au moyen d'un tableur, en l'occurrence Excel. Tout est transposable dans d'autres programmes.


Pour l'uniformisation des différents fichiers d'éphémérides, chacun comprend

- une feuille d'éphémérides pour chaque objet ([Lune] et [Soleil])
- une feuille de calculs et de graphiques ([Eclipse]).

Données IMCCE (<http://www.imcce.fr>)

Les pages de résultats des éphémérides de l'IMCCE comportent une partie mise en page encadrant les résultats. Les données utiles commencent à la première ligne de « # » et finissent à la dernière ligne d'éphémérides.

Partie à sélectionner
à copier, coller
dans les feuilles Excel



INSTITUT DE MÉCANIQUE CÉLESTE ET DE CALCUL DES ÉPHÉMÉRIDES

EPHEMERIDES -> RESULTATS

• Senseur d'éphémérides de l'IMCCE • Signification des paramètres calculés
• Fichier résultats au format asii

EPHEMERIDES DES CORPS DU SYSTEME SOLAIRE

Satellite 10 Lune
Theorie planétaire DE405/LE405
Coordonnées Astronomiques J2000
Centre du repere : geocentre
Coordonnées ecliptiques (lambda, beta)

Date UTC	h	m	s	o	Long.	o	Lat.	o	Distance Rt.	V.Mag	Phase	Elong.
3 3 2007 20 0 0.00	161	14	23.0806	+00 26 31.2304	63.013868989	-12.52	1.58	178.41				
3 3 2007 20 0 30.00	161	14	36.1491	+00 26 29.8401	63.013985541	-12.52	1.58	178.42				
3 3 2007 20 1 0.00	161	14	53.2175	+00 26 28.4497	63.014102086	-12.52	1.57	178.42				
3 3 2007 20 1 30.00	161	15	8.2858	+00 26 27.0594	63.014218622	-12.52	1.57	178.43				
3 3 2007 20 2 0.00	161	15	23.3541	+00 26 25.6691	63.014335149	-12.52	1.57	178.43				
3 3 2007 20 2 30.00	161	15	38.4223	+00 26 24.2787	63.014451667	-12.52	1.56	178.43				
3 3 2007 20 3 0.00	161	15	53.4905	+00 26 22.8884	63.014568177	-12.52	1.56	178.44				
3 3 2007 20 3 30.00	161	16	8.5586	+00 26 21.4980	63.014684678	-12.53	1.56	178.44				
3 3 2007 20 4 0.00	161	16	23.6266	+00 26 20.1077	63.014801170	-12.53	1.55	178.44				
3 3 2007 20 4 30.00	161	16	38.6946	+00 26 18.7173	63.014917653	-12.53	1.55	178.45				

Remplir et soumettre le formulaire pour calculer une nouvelle éphéméride:

Chargez un fichier de dates...

ou

Année: 2007 Mois: 03 Jour: 03 Heure: 20 Minutes: Secondes: Date courante Effacer

• Nombre de dates: 1000 (+ 5000)

• Pas de calcul: 30 secondes

• Format des dates?: Date gregorienne

• Format des sorties?: asii

Attention: Ce senseur d'éphémérides a été mis en place avec l'aide du Ministère de l'Éducation Nationale, du CNRS et du CNES. Si vous devez utiliser ces données en vue d'une diffusion publique, veuillez nous consulter

retour haut de page |
imprimer la page |
paramétrer le paramètre |

Pour simplifier les fichiers, il n'y a pas de liens entre les feuilles d'éphémérides de chaque corps, et les données de la feuille de calculs. Libre à l'utilisateur de le faire.

En conséquence, les éphémérides de la feuille « calculs et graphiques » sont à prendre dans les feuilles des corps et à reporter dans la feuille de « calculs graphique » avec « copié-collé spécial valeurs ».

Calcul d'éphémérides à l'IMCCE

Pour la Lune et le Soleil :

- ① Un corps du Système solaire Lune, Soleil
- ② La théorie planétaire garder celle par défaut
- ③ Centre du repère géocentrique
- ④ Plan de référence écliptique
- ⑤ Type de coordonnées sphériques
- ⑥ Type d'éphémérides Astrométrique J2000
- ⑦ L'échelle de temps UTC
- ⑧ L'époque du calcul
pas échantillonnage 2007 03 03 21 00 00
nombre de points 30s
 600
- ⑨ Lancer le calcul

Sauvegarde des résultats

Ouvrir Excel et créer 3 feuilles que l'on renommera :

- [Eclipse]
- [Lune]
- [Soleil]

Pour la Lune et Soleil, faire un copié-collé des données obtenues sur le site de l'IMCCE dans la colonne A de la feuille correspondante Excel.

Pour la partie à sélectionner voir page précédente.

Note : si dans Excel, le caractère décimal des feuilles de calcul n'est pas le point décimal vous pouvez soit

- dans Excel mettre le point comme caractère décimal,
- dans les éphémérides changer les points en virgules.

Page web à remplir pour le calcul

Conversion des données en valeurs numériques

Les formules de calcul s'appliquent aux cellules de la ligne 16. Il suffit ensuite d'appliquer par *sélectionner glisser* pour l'appliquer aux autres lignes de données, ce qui incrémente le rang des cellules dans les formules.

> Dates et heures ([colonne B et C])

Dates : formule [cellule B16]

=DATE(CTXT(STXT(A16;9;4));CTXT(STXT(A16;4;3));CTXT(STXT(A16;1;3)))

Heure : formule cellule C16

=(CNUM(STXT(A16;14;2))+CNUM(STXT(A16;17;2))/60+CNUM(STXT(A16;20;5))/3600)/24

Ne pas oublier de mettre le format date dans la *colonne B* et format Heure dans la *colonne C*.

Les dates étant communes aux deux objets, il suffit de les extraire une fois pour la Lune ([colonnes B et C]) et les reporter dans les *colonnes A et B* de la "feuille Eclipse" par un **collé spécial valeurs**.

Echelle des temps pour les graphes et les calculs des instants

Lorsque l'éclipse commence le soir et fini le lendemain matin, il y a un saut de 24 à 0 dans les valeurs de l'heure, ce qui est très gênant pour tracer un graphe avec le temps comme abscisses.

Sous Excel les heures sont codées en jours décimaux (0h ⇒ 0.0, 1h ⇒ 0.041666, 2h ⇒ 0.08333, 3h ⇒ 0.25 etc, mais apparaissent sous la forme conventionnelle hh:mm:ss par le formatage.

Pour éviter le problème au passage à 0h, on ajoute 1 (24h).

Dans la [feuille Eclipse] on crée une [colonne C] heure continue :

- commence le 3 au soir (20h, 21h...)

- se poursuit le 4 mars par 24h, 25h, 26h...

mais s'affichera 0h, 1h, 2h... avec le format heure d'Excel.

Ceci se traduit par la formule de la [cellule C16] : =(A16-\$A\$16+B16), à reporter dans toutes les cellules de la [colonne C]. Cette formule permet de travailler en continu sur plusieurs jours si nécessaire.

> Longitudes et latitudes écliptiques de la Lune ([feuille Lune colonnes D et E])

Longitudes écliptiques de la Lune

=CNUM(STXT(A16;27;3))+CNUM(STXT(A16;31;2))/60+CNUM(STXT(A16;34;7))/3600

Latitudes écliptiques de la Lune

=(CNUM(STXT(A16;44;2))+CNUM(STXT(A16;47;2))/60+CNUM(STXT(A16;50;7))/3600)*
(SI(STXT(A16;43;1)="-";-1;1))

> Longitudes et latitudes écliptiques de la direction de l'axe du cône d'ombre [feuille Soleil col. D et E]

Longitudes écliptiques de la direction du cône d'ombre (opposée à la direction du Soleil)

=MOD(CNUM(STXT(A16;27;3))+CNUM(STXT(A16;31;2))/60+CNUM(STXT(A16;34;7))/3600+180;360)

Latitudes écliptiques de la direction du cône d'ombre

=-((CNUM(STXT(A16;44;2))+CNUM(STXT(A16;47;2))/60+CNUM(STXT(A16;50;7))/3600)*
(SI(STXT(A16;43;1)="-";-1;1)))

> Distances

Terre-Lune [Feuille Lune Colonne F cellule F17] =CNUM(STXT(A17;60;11))*\$B\$6

Terre Soleil [Feuille Soleil Colonne F cellule F17] =CNUM(STXT(A17;60;11))*\$B\$10

Données générales

Mettre dans les cellules de la feuille éclipse :

- F3 : le rayon de la Lune 1737.4 km
- F5 : le rayon équatorial de la Terre 6378.14 km
- F7 : le rayon du Soleil 696000 km
- F9 : l'unité astronomique 149597870 km

Dans les formules, pour leur utilisation, il faudra les écrire avec \$: \$C\$L pour éviter les décalages

Disposition feuille Lune et Soleil – Données tirées des éphémérides

Colonne	Feuille Lune	Feuille Soleil
A	Données IMCCE	Données IMCCE
B	Dates	Dates
C	heures	heures
D	Longitude écliptique axe du cône d'ombre	Longitude écliptique
E	Latitude écliptique axe du cône d'ombre	Latitude écliptique
F	Distance (rayons équatoriaux terrestres)	Distance (unités astronomiques)

[Feuille Lune] :

A													B	C	D	E	F				
EPHEMERIDES DES CORPS DU SYSTEME SOLAIRE																					
Satellite 10 Lune																					
Theorie planetaire DE405/LE405																					
Coordonnees Astrometriques J2000																					
Centre du repere : geocentre																					
Coordonnees ecliptiques (lambda, beta)																					
Date UTC													Date	Heure	Longitude	Latitude	Distance				
h m s o ' "															(°)	(°)	(R Terre)				
16	3	3	2007	20	0	0.00	161	14	23.0806	+00	26	31.2304	63.013868988	-12.52	1.58	178.41	03/03/2007	20:00:00	161.239745	0.44201	63.01387
17	3	3	2007	20	0	30.00	161	14	38.1491	+00	26	29.8401	63.013985541	-12.52	1.58	178.42	03/03/2007	20:00:30	161.243930	0.44162	63.01399
18	3	3	2007	20	1	0.00	161	14	53.2175	+00	26	28.4497	63.014102086	-12.52	1.57	178.42	03/03/2007	20:01:00	161.248116	0.44124	63.01410

Disposition feuille Eclipse – Données tirées des éphémérides

Colonne	Contenu	Provenance par Copié-collé spécial valeurs
A	Dates	[Feuille Lune col. B]
B	heures	[Feuille Lune col. C]
D	Longitude écliptique axe du cône d'ombre	[Feuille Soleil col. D]
E	Latitude écliptique axe du cône d'ombre	[Feuille Soleil col. E]
F	Distance Soleil (unités astronomiques)	[Feuille Soleil col. F]
G	Longitude écliptique Lune	[Feuille Lune col. D]
H	Latitude écliptique Lune	[Feuille Lune col. E]
I	Distance Lune (rayons équatoriaux terrestres)	[Feuille Lune col. F]

A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U
Eclipse totale de Lune																				
Le 3 mars 2007																				
Ephémérides IMCCE																				
					R Lune															
					1747.40															
					Rayon Terre															
					6378.14															
					R Soleil															
					696000.00															
					1 UA															
					149597871															
Date	Heure	Heure graphique	Direction cône d'ombre		Distance Soleil	Lune		ΔL	ΔI	θ	Lune	D Soleil	D Lune	Cône d'Ombre alpha	Rayon du cône d'ombre à la hauteur de la Lune			HL	R ₁ cosα	
			Longitude (°)	Latitude (°)	(u.a.)	Longitude (°)	Latitude (°)	(R Terre)	(min d'arc)	(min d'arc)	(min d'arc)	rayon ang. (°)	km	km	alpha (°)	R Ombre (°)	R Ombre + R Lune	R Ombre - R Lune	(km)	(km)
03/03/2007	21:00:00	21:00:00	162.7989524	-0.0002173	0.9913639	161.7419128	0.3966578	63.02779	-63.42237	23.753	67.72381	14.94298	148305934.4	402000.1	0.2664261	38.56	53.49921	23.61324	3409.66	1747.4
03/03/2007	21:00:30	21:00:30	162.7993004	-0.0002173	0.9913640	161.7460966	0.3952714	63.02791	-63.19223	23.729	67.50018	14.94296	148305947.9	402000.8	0.2664260	38.56	53.49908	23.61317	3383.53	1747.4
03/03/2007	21:01:00	21:01:00	162.7996484	-0.0002173	0.9913641	161.7502804	0.3948851	63.02802	-62.96208	23.706	67.27661	14.94293	148305961.3	402001.6	0.2664260	38.56	53.49896	23.61310	3357.42	1747.4

Disposition feuille Eclipse – Données calculées à partir des éphémérides

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	
1	Eclipse totale de Lune																					
2	Le 3 mars 2007																					
3	Ephémérides IMCCE																					
4						R Lune																
5						1747.40																
6						Rayon Terre																
7						6378.14																
8						R Soleil																
9						696000.00																
10						1 UA																
11						149597871																
12			Direction cône d'ombre			Distance	Lune			Lune			Cône d'ombre			Rayon du cône d'ombre à la hauteur de la Lune			HL	R ₁ cosα		
13	Date	Heure	Heure graphique	Longitude (°)	Latitude (°)	Soleil (u.a.)	Longitude (°)	Latitude (°)	Distance (R Terre)	ΔL (min d'arc)	ΔI (min d'arc)	θ (min d'arc)	rayon ang. (°)	D Soleil km	D Lune km	α (°)	R Ombre (°)	R Ombre + R Lune (°)	R Ombre - R Lune (°)	(km)	(km)	
14																						
15																						
16	03/03/2007	21:00:00	21:00:00	162.7989624	-0.0002173	0.9913639	161.7419128	0.3956578	63.02778	-63.42237	23.753	67.72381	14.94298	148305934.4	402000.1	0.2664261	38.56	53.49921	23.61324	3409.66	1747.4	
17	03/03/2007	21:00:30	21:00:30	162.7993004	-0.0002173	0.9913640	161.7460966	0.3952714	63.02791	-63.19223	23.729	67.50018	14.94296	148305947.9	402000.8	0.2664260	38.56	53.49908	23.61317	3383.53	1747.4	
18	03/03/2007	21:01:00	21:01:00	162.7996484	-0.0002173	0.9913641	161.7502804	0.3948851	63.02802	-62.96208	23.706	67.27661	14.94293	148305961.3	402001.6	0.2664260	38.56	53.49896	23.61310	3357.42	1747.4	


Contenu des cellules des colonnes C, et J à U

Col.	Contenu	Remarque
C	heures continues	pour le changement de jour
J	Δ Longitudes de l'axe du cône d'ombre	en minutes d'arc
K	Δ Latitude de l'axe du cône d'ombre	en minutes d'arc
L	θ distance angulaire, axe du cône - Lune	en minutes d'arc
M	ρ _{Lune} diamètre angulaire de la Lune	en minutes d'arc
N	D _{Soleil} distance Terre Soleil	en kilomètres
O	D _{Lune} distance Terre Lune	en kilomètres
P	α demi angle d'ouverture du cône d'ombre	en degrés
Q	ρ _{Ombre} Rayon angulaire de l'ombre à la hauteur de la Lune	en minutes d'arc
R	Angle ρ _{ombre} + ρ _{Lune}	en minutes d'arc
S	Angle ρ _{ombre} - ρ _{Lune}	en minutes d'arc
T	OL Rayon du cône d'ombre à la hauteur de la Lune	en kilomètres
U	HL Distance centre Lune bord de l'ombre à la hauteur de la Lune	en kilomètres

Que peut-on calculer et tracer ?

De la Terre on va donc voir passer la Lune dans son ombre. Faisons l'état des positions et instants remarquables et calculables. La Lune se rapproche du cône d'ombre,

- entre en contact avec lui (position L₁, instant t₁)
- la Lune s'obscurcit et arrive l'instant où elle est entièrement dans l'ombre (instant t₂, position L₂)
- elle passe au plus près du centre du cône, au maximum de l'éclipse, (instant t_M, position L_M)
- elle va de nouveau émerger de l'ombre (instant t₃, position L₃)
- et va quitter définitivement l'ombre de la Terre (instant t₄, position L₄)



Quand ont lieu ces instants ?

1) Maximum

Il suffit de calculer à l'aide des éphémérides l'angle entre les directions Lune et l'axe du cône et de chercher l'instant du minimum.

Nous avons dans les [colonnes J] et [K] les différences de longitude et de latitude.

L'angle θ (angle *OTL*) se calcule de deux façons :

a) De façon approchée :

$$\theta = \sqrt{(\Delta L \cos l)^2 + \Delta l^2}$$


car les angle ΔL et Δl sont petits et ici $l = 0$ (sur l'écliptique).

b) De façon rigoureuse, dans le triangle sphérique *OHL*, on a :

$$\cos \theta = \cos \Delta L \cdot \cos \Delta l$$

Le calcul de θ se fait dans la [colonne L] que l'on exprimera en minutes d'arc.

Formule de calcul : = DEGRES(ACOS(COS(RADIANS(J16/60))*COS(RADIANS(L16/60))))*60
ou = RACINE(K16^2+L16^2)




Graphique

Construire dans la [feuille éclipse] : variation de θ en fonction du temps UTC.

On ajustera les paramètres de temps (à rentrer en jours décimaux) pour une lecture facile (heures entières et demi-heures)

Instant du maximum

Avec un pas d'échantillonnage de 30s, et au vu des variations, le calcul de l'instant du maximum de l'éclipse (minimum de θ) se fera par l'ajustement d'une parabole à partir des trois points les plus proche du minimum.



Le temps du maximum est déterminable approximativement dans la colonne M (minimum) et sur le graphique.


Pour préciser l'instant du maximum, on va calculer l'ajustement d'une parabole passant par les trois points encadrant le minimum de θ .

Recherche de l'extremum d'une parabole passant par trois points

Si le pas d'échantillonnage est constant $\Delta t = (t_b - t_c) = (t_c - t_b)$ (30s) :

$$t_{min} = -\frac{1}{2} \cdot \frac{x_c - x_a}{x_c + x_a - 2x_b} \Delta t + t_b$$

Voir fichier *data_ephemerides.pdf* page 12.



Formule : = -(L300 - L298)/(L300+L298 - 2*L299)/2*(C299 - C298)+C299
où a, b et c correspondent aux rangs 298, 299 et 300.

Pour continuer les calculs, il nous faut en fonction du temps :

- la distance Terre Soleil en kilomètres ([*colonne M*]) :

$$D_S = D_S(\text{u.a.}) \cdot 1_{\text{u.a.}} (\text{km})$$

formule : =F16*\$F\$9

- la distance Terre Lune en kilomètres ([*colonne N*]) :

$$D_L = D_L(\text{en rayon terrestre}) \cdot R_T (\text{km})$$

formule : =I16*\$F\$5

- le rayon angulaire de la Lune ([*colonne O*]) exprimé en minutes d'arc :

$$\rho_L = \tan^{-1} \frac{R_{\text{Lune}}}{D_{\text{Lune}}}$$

formule : =DEGRES(ATAN(\$F\$3/(N16)))*60

Pour calculer la grandeur de l'ombre ([*colonne P*]), il faut l'angle α de demi-ouverture du cône, qui par simple géométrie s'écrit (figure page 6) :

$$\alpha = \sin^{-1} \frac{R_S - R_T}{D_S}$$

formule : =DEGRES(ASIN((\$F\$7-\$F\$5)/N16))

2) Instants des contacts

En première approximation, on peut estimer à quelques minutes près les moments des contacts :

1) entrée et sortie de la Lune de l'ombre

$$\theta_1 = OL_1 = \rho_{\text{ombre}} + \rho_{\text{Lune}} \quad [\text{colonne R}]$$

$$\theta_4 = OL_4 = \rho_{\text{ombre}} + \rho_{\text{Lune}} \quad [\text{colonne R}]$$

2) entrée et sortie de la totalité

$$\theta_2 = OL_2 = \rho_{\text{ombre}} - \rho_{\text{Lune}} \quad [\text{colonne S}]$$

$$\theta_3 = OL_3 = \rho_{\text{ombre}} - \rho_{\text{Lune}} \quad [\text{colonne S}]$$

Les rayons et distances sont alors des distances angulaires.

Il faut donc calculer en fonction du temps :

- le rayon angulaire de l'ombre à la distance de la Lune (démonstration en annexe)

$$\rho_{\text{ombre}} = \tan^{-1} \left(\frac{R_T}{\cos \alpha D_L \cos \theta} - \tan \alpha \right)$$

- le rayon angulaire de la Lune

et chercher les instants où θ vaut soit :

$$\theta_1 \text{ et } \theta_4 \quad \rho_+ = \rho_{\text{ombre}} + \rho_{\text{Lune}} \quad (\text{entrée et sortie de l'ombre})$$

$$\theta_2 \text{ et } \theta_3 \quad \rho_- = \rho_{\text{ombre}} - \rho_{\text{Lune}} \quad (\text{début et fin de la totalité})$$

Dans la feuille de calcul nous aurons les colonnes

[*colonne Q*] rayon angulaire de l'ombre (en min d'arc)

$$= \text{DEGRES}(\text{ATAN}(\frac{F5}{\text{COS}(\text{RADIANS}(P16)) * O16 * \text{COS}(\text{RADIANS}(L16/60))}) - \text{TAN}(\text{RADIANS}(P16)))) * 60$$

[*colonne S*] angle formule : =Q16+O16

[*colonne T*] angle $\rho_{\text{ombre}} - \rho_{\text{Lune}}$ formule : =Q16-O16

Les variations de θ , $(\rho_{\text{ombre}} + \rho_{\text{Lune}})$ et $(\rho_{\text{ombre}} - \rho_{\text{Lune}})$ sur de courts instants sont quasi linéaires.




On recherchera donc le moment de contact par le calcul de l'intersection de deux droites. L'instant cherché sera l'abscisse de l'intersection.

• Equations des deux droites

$$\frac{y - y_a}{t - t_a} = \frac{y_b - y_a}{t_b - t_a} \quad \text{et} \quad \frac{z - z_a}{t - t_a} = \frac{z_b - z_a}{t_b - t_a}$$

• Abscisse de l'intersection

$$t = (t_b - t_a) \frac{z_a - y_a}{y_b - y_a - z_b + z_a} + t_a$$



1) entrée et sortie de l'ombre

Comparaison des [colonnes L] (θ) et [S] ($\rho_{\text{ombre}} + \rho_{\text{Lune}}$)

formule à l'intersection : $=(C201-C200)*(X200-V200)/(V201-V200-X200+X200)+C200$

2) début et fin de la totalité

Comparaison des [colonnes L] (θ) et [T] ($\rho_{\text{ombre}} - \rho_{\text{Lune}}$)

formule à l'intersection : $=(C350-C349)*(V349-Y349)/(Y350-Y349-V350+V349)+C349$


Grandeur de l'éclipse

Au maximum, la grandeur de l'éclipse s'exprime par la relation

$$g = \frac{AB}{AC}$$

Si g est < 1 , l'éclipse est partielle
 ≥ 1 , l'éclipse est totale et d'autant plus longue que g est grand.

Maximum du l'éclipse



$$\begin{aligned} AL &= LC = \rho_{\text{Lune}} \\ OB &= \rho_{\text{ombre}} \\ OL &= \theta_{\text{max}} \\ AB &= AL - OL + OB \end{aligned}$$

$$g = \frac{AL - OL + OB}{AC} = \frac{\rho_{\text{Lune}} - \theta + \rho_{\text{ombre}}}{2\rho_{\text{Lune}}} \quad \text{formule } =(O299-L299+Q299)/(2*O299)$$


Tracés

Les différents instants peuvent être réunis dans un seul graphe où l'on porte en fonction du temps:

$\Delta L, \Delta l, \theta, \rho_{\text{ombre}}, \rho_{\text{ombre}} + \rho_{\text{Lune}}, \rho_{\text{ombre}} - \rho_{\text{Lune}}$

Le minimum de θ donne l'instant du maximum.


Les intersections de θ avec $\rho_{\text{ombre}} + \rho_{\text{Lune}}, \rho_{\text{ombre}} - \rho_{\text{Lune}}$ donnent les instants de début et sortie de l'ombre, le début et la fin de l'éclipse.



Un peu plus de mathématiques

La comparaison de rayons angulaires n'est qu'une approximation pour les instants de contact.

Pour être plus rigoureux, on va calculer le rayon de l'ombre à la hauteur de la Lune et le comparer à la distance réelle de la Lune à l'axe du cône d'ombre.



Un peu de géométrie :

A tout instant, les éphémérides permettent de calculer $R_{\text{ombre}} = OH$, le rayon du cône d'ombre à la hauteur de la Lune et la distance de la Lune à l'axe du cône d'ombre.

$$OH = OC \sin \alpha = (TC - TO) \tan \alpha$$

$$TO = D_L \cos \theta$$

$$TC = R_T \sin \alpha$$

$$OH = (R_T \sin \alpha - D_L \cos \theta) \tan \alpha$$

$$OL = D_L \sin \theta$$

Remarque : le rayon angulaire de l'ombre s'en déduit : $\rho_{\text{ombre}} = \tan^{-1} \frac{OH}{TO} = \frac{OH}{D_L \cos \theta}$

On en déduit HL .

Quand la Lune est en L_1, L_2, L_3 , et L_4 : $|HL| = \frac{R_L}{\cos \alpha}$

Avant le premier contact, $|HL| > R_L / \cos \alpha$

Au contact $|HL| = R_L / \cos \alpha$

Après le premier contact $|HL| < R_L / \cos \alpha$

Dans le tableur nous allons calculer pour chaque instant des éphémérides

[colonne T] $|LH| = |OL - OH| = D_L \sin \theta - (R_T \sin \alpha - D_L \cos \theta) \tan \alpha$

Formule : `=ABS(O16*SIN(RADIANS(L16/60))-(F$5/SIN(RADIANS(P16))-O16*COS(RADIANS(L16/60)))*TAN(RADIANS(P16)))`

[colonne U] $R_L / \cos \alpha$ Formule : `=F$3/COS(RADIANS(Q16))`

On recherchera les instants d'égalité de HL et $R_L / \cos \alpha$ comme précédemment par l'intersection de deux droites.

Formule du calcul des instants :


$$=(C81-C80)*(T80-U80)/(U81-U80-T81+T80)+C80$$

si l'égalité se produit entre les rangs 80 et 81.

L'intersection peut être tracée en portant dans un graphe les deux segments de droites

- les [cellules C79 à C82] en abscisse et [T79 à T82] en ordonnées
- les [cellules C79 à C82] en abscisse et [U79 à U82] en ordonnées

$$\text{Grandeur de l'éclipse : } g = \frac{R_{\text{Lune}} - OL + R_{\text{ombre}}}{2R_{\text{Lune}}}$$



Feuille Eclipse – formules (récapitulatif)

Les formules seront explicités dans la suite

Col.	Contenu	Formule
<i>C</i>	heures	= $(A16 - \$A\$16 + B16)$
<i>J</i>	Δ Longitudes	= $(G16 - D16) * 60$
<i>K</i>	Δ Latitude	= $(E16 - H16) * 60$
<i>L</i>	θ	= $DEGRES(ACOS(COS(RADIANS(J16/60)) * COS(RADIANS(K16/60)))) * 60$
<i>M</i>	R_{Lune}	= $DEGRES(ATAN(\$F\$3 / (I16 * \$F\$5))) * 60$
<i>N</i>	D_{Soleil}	= $F16 * \$F\9
<i>O</i>	D_{Lune}	= $I16 * \$F\5
<i>P</i>	α	= $DEGRES(ASIN((\$F\$7 - \$F\$5) / N16))$
<i>Q</i>	Rombre	= $DEGRES(ATAN(\$F\$5 / (COS(RADIANS(P16)) * O16 * COS(RADIANS(L16/60))) - TAN(RADIANS(P16)))) * 60$
<i>R</i>	Angle $\rho_{ombre} + \rho_{Lune}$	= $Q16 + M16$
<i>S</i>	Angle $\rho_{ombre} - \rho_{Lune}$	= $Q16 - M16$
<i>T</i>	<i>HL</i>	= $ABS(O16 * SIN(RADIANS(L16/60)) - (\$F\$5 / SIN(RADIANS(P16))) - O16 * COS(RADIANS(L16/60)) * TAN(RADIANS(P16)))$
<i>U</i>	$R_{Lune} / \cos \alpha$	= $\$F\$3 / COS(RADIANS(P16))$

Résultats

Phénomènes	Temps TU	IMCCE (UTC)
début éclipse par l'ombre	21:32:06	21h 30.2m
début totalité	22:46:50	22h 44.0m
maximum de l'éclipse	23:21:08	23h 20m 44.531s
fin totalité	23:56:24	23h 57.4m
Fin éclipse par l'ombre	1:11:08	1h 11.3m
grandeur	1.212	1.2330754

Autres données IMCCE

Contacts Angles

Premier contact extérieur avec l'ombre	137.0°
Premier contact intérieur avec l'ombre	343.8°
Dernier contact intérieur avec l'ombre	72.7°
Denier contact extérieur avec l'ombre	279.6°

Maximum de l'éclipse	le 3 mars 2007 à 23h 20m 44.531s UTC
Ascension droite du centre de l'ombre	10h 57m 19.17s
Déclinaison du centre de l'ombre	+ 6° 40' 46.35"
Diamètre du cône d'ombre	+ 78.17' (trouvé 77.06')
Diamètre du cône de pénombre	+142.70'
Parallaxe équatoriale du Soleil	+ 8.87"
Ascension droite du centre de la Lune	10h 57m 52.19s
Déclinaison du centre de la Lune	+ 6° 56' 0.71"
Diamètre apparent de la Lune	29.71' 29.69
Parallaxe équatoriale de la Lune	+ 0° 54' 31.06"